



BUGS Kidney: Weibull regression with random effects

McGilchrist and Aisbett (1991) analyse time to first and second recurrence of infection in kidney patients on dialysis using a Cox model with a multiplicative frailty parameter for each individual. The risk variables considered are age, sex and underlying disease (coded other, GN, AN and PKD). A portion of the data are shown below.

Patient Number	Recurrence time t	Event (2 = cens)	Age at time t	Sex (1 = female)	Disease (0 = other; 1 = GN 2 = AN; 3 = PKD)
1	8,16	1,1	28,28	0	0
2	23,13	1,2	48,48	1	1
3	22,28	1,1	32,32	0	0
4	447,318	1,1	31,32	1	0
....					
35	119,8	1,1	22,22	1	1
36	54,16	2,2	42,42	1	1
37	6,78	2,1	52,52	1	3
38	63,8	1,2	60,60	0	3

We have analysed the same data assuming a parametric Weibull distribution for the survivor function, and including an additive random effect b_i for each patient in the exponent of the hazard model as follows

$$t_{ij} \sim \text{Weibull}(r, \mu_{ij}) \quad i = 1, \dots, 38; \quad j = 1, 2$$

$$\log \mu_{ij} = \alpha + \beta_{age} \text{AGE}_{ij} + \beta_{sex} \text{SEX}_i + \beta_{disease1} \text{DISEASE}_{i1} + \\ \beta_{disease2} \text{DISEASE}_{i2} + \beta_{disease3} \text{DISEASE}_{i3} + b_i$$

$$b_i \sim \text{Normal}(0, \tau)$$

where AGE_{ij} is a continuous covariate, SEX_i is a 2-level factor and DISEASE_{ik} ($k = 1, 2, 3$) are dummy variables representing the 4-level factor for underlying disease. Note that the survival distribution is a truncated Weibull for censored observations as discussed in the mice example. The regression coefficients and the precision of the random effects τ are given independent "non-informative" priors, namely


$$b_k \sim \text{Normal}(0, 0.0001)$$

$$\tau \sim \text{Gamma}(0.0001, 0.0001)$$

The shape parameter of the survival distribution r is given a $\text{Gamma}(1, 0.0001)$ prior which is slowly decreasing on the positive real line.

The graphical model and BUGS language are given below.

Graphical model for kidney example:



BUGS language for kidney example

```

model
{
  for (i in 1:N) {
    for (j in 1:M) {
      # Survival times bounded below by censoring times:
      t[i,j] ~ dweib(r,mu[i,j]) I(t.cen[i,j],);
      log(mu[i,j]) <- alpha + beta.age*age[i,j]
                    + beta.sex *sex[i]
                    + beta.dis[disease[i]] + b[i];
    }
    # Random effects:
    b[i] ~ dnorm(0.0, tau)
  }
  # Priors:
  alpha ~ dnorm(0.0, 0.0001);
  beta.age ~ dnorm(0.0, 0.0001);
  beta.sex ~ dnorm(0.0, 0.0001);
  # beta.dis[1] <- 0; # corner-point constraint
  for(k in 2 : 4) {
    beta.dis[k] ~ dnorm(0.0, 0.0001);
  }
  tau ~ dgamma(1.0E-3, 1.0E-3);
  r ~ dgamma(1.0, 1.0E-3);
  sigma <- 1/sqrt(tau); # s.d. of random effects
}

```



```

0, 0,
0, 0,
0, 0,
0, 0,
5, 0,
0, 0,
0, 5,
0, 0,
54, 16,
6, 0,
0, 8), .Dim = c(38, 2)),
age = structure(
.Data = c(28, 28,
48, 48,
32, 32,
31, 32,
10, 10,
16, 17,
51, 51,
55, 56,
69, 69,
51, 52,
44, 44,
34, 34,
35, 35,
42, 42,
17, 17,
60, 60,
60, 60,
43, 44,
53, 53,
44, 44,
46, 47,
30, 30,
62, 63,
42, 43,
43, 43,
57, 58,
10, 10,
52, 52,
53, 53,
54, 54,
56, 56,
50, 51,
57, 57,
44, 45,
22, 22,
42, 42,
52, 52,
60, 60), .Dim = c(38, 2)),
beta.dis = c(0, NA, NA, NA),
sex = c(0, 1, 0, 1, 0, 1, 0, 1, 1, 0, 1, 1, 1, 1, 1, 1,
0, 1, 1, 1, 0, 1, 1, 1, 0, 1, 1, 1, 1, 1, 1, 1, 1, 1, 0),
disease = c(1, 2, 1, 1, 1, 1, 1, 2, 2, 3, 2, 3, 1, 3, 1, 3, 1, 1, 2, 1, 4, 1, 3, 3,
3, 3, 2, 3, 2, 3, 3, 4, 2, 1, 1, 4, 4)) $\hat{}$ 

```

Inits \Rightarrow list(beta.age = 0, beta.sex = 0, beta.dis=c(NA,0,0,0),
alpha = 0, r=1, tau=0.3) $\hat{}$

Results

A 1000 update burn in followed by a further 10000 updates gave the parameter estimates

node	mean	sd	MC error	2.5%	median	97.5%	start	sample	
alpha	-4.6	0.8962	0.07002	-6.541	-4.541	-3.065	1001	10000	
beta.age	0.003027	0.01475	9.703E-4	-0.02411	0.002423	0.03368	1001	10000	
beta.dis[2]	0.1329	0.5393	0.02148	-0.9443	0.1276	1.237	1001	10000	
beta.dis[3]	0.6444	0.5301	0.02364	-0.4158	0.6428	1.711	1001	10000	
beta.dis[4]	-1.168	0.8335	0.03382	-2.772	-1.182	0.5535	1001	10000	
beta.sex	-1.938	0.4854	0.02524	-2.952	-1.917	-1.033	1001	10000	
r	1.215	0.1623	0.01293	0.9275	1.204	1.536	1001	10000	
sigma	0.6374	0.357	0.02847	0.04597	0.653	1.322	1001	10000	